

## 5.17 モンテカルロシミュレーション MCMC

ver.0.1

ここでは、当初開発された離散ジョイント分布による Gibbs Sampling から始めて、連続分布によるもの、そして、Metropolis-Hastings の方法、その Bivariate のケース、そして Metropolis-within-Gibbs の方法へと発展させてプログラムをしていきます。

### Gibbs Sampling

離散でも連続でもどちらでもかまいませんが、ある分布  $D$  のもとに、 $n$  個のパラメータからなる をシミュレーションすることにしましょう。今、初期値が仮に

$$x_1^{(0)}=k_1, \quad x_2^{(0)}=k_2, \dots, \quad x_n^{(0)}=k_n$$

と  $D$  の範囲内にある一定の値に決まっているものとします。ここから、

$$\begin{aligned} x_1^{(k)} &\sim p(x_1 | x_2^{(k-1)}, x_3^{(k-1)}, \dots, x_n^{(k-1)}, D) \\ x_2^{(k)} &\sim p(x_2 | x_1^{(k)}, x_3^{(k-1)}, \dots, x_n^{(k-1)}, D) \\ &\vdots \\ x_{n-1}^{(k)} &\sim p(x_{n-1} | x_1^{(k)}, x_2^{(k)}, \dots, x_{n-2}^{(k)}, x_n^{(k-1)}, D) \\ x_n^{(k)} &\sim p(x_n | x_1^{(k)}, x_2^{(k)}, \dots, x_{n-2}^{(k)}, x_{n-1}^{(k)}, D) \end{aligned}$$

上のような上付きの  $k$  というステージを  $k = 1$  から times 番目まで times 回繰り返して、

$$x = (x_1, x_2, \dots, x_n)$$

をシミュレーションにより求めます。これが Gibbs Sampling です。これを 2 つのパラメータからなる離散分布にもとづくもので考えたのが下の例です。すなわち、初期を仮に  $(1,1)$  から始めるために、 $D$  の分布の範囲内の  $x_1^{(0)} = 1, \quad x_2^{(0)} = 1$  と止めておいて、

$$\begin{aligned} x_1^{(k)} &\sim p(x_1 | x_2^{(k-1)}, D) \\ x_2^{(k)} &\sim p(x_2 | x_1^{(k)}, D) \end{aligned}$$

の第  $k$  ステップを逐次的に times 回繰り返すものです。  $x_1 = \{1, 2\}$ ,  $x_2 = \{1, 2, 3\}$  として、離散 Joint 分布

$$\begin{bmatrix} 0.1 & 0.2 & 0.2 \\ 0.1 & 0.1 & 0.3 \end{bmatrix}$$

という確率分布でこれらのパラメータが出てくるものとします。下の例では、 $x_1$  と  $x_2$  がそれぞれ離散 Joint 分布の行と列のインデックス番号  $a$  と  $b$  とします。初期値  $(a,b)=(1,1)$  から ( $D$  は 0 を取らないので 0 と置いてはいけません) それぞれの列それに行に注目して Conditional 分布を計算して、その確率にもとづいて逐次的に  $a$  を、そしてその当期の  $a$  を用いて (その行に注目して Conditional 分布を求めて)  $b$  を求めています。プログラムでは `usample(x,p)` に対して、 $x = \{1,2\}$  を前期の  $b$  列に注目した Conditional 分布から取りだし、 $a$  として、さらにその当期の  $a$  を使って、 $x=\{1,2,3\}$  を  $a$  行の Conditional 分布から取り出す計算をするものが、

```
a=usample(seqa(1,1,rows(joint)),joint[:,b]/sumc(joint[:,b]));
b=usample(seqa(1,1,cols(joint)),joint[a,:]/sumc(joint[a,:]));
```

という部分になっています。つまり、{1,2}および{1,2,3}の中から、それぞれの列または行にもとづいた Conditional 分布確率でサンプリングしています。その際、最初のサンプリングではbは前期のもの、第2式のサンプリングでは、第1式で得られたその期のaにもとづいてサンプリングが行なわれます。このステージを初期値のaとbを固定してやってから、times 回繰り返してaとbの出方を調べて、ma および mb という列ベクトルに格納していきます。これらの列ベクトルが Gibbs Sampling によるシミュレーション結果です。

プログラム

```
new; cls;
rndseed 1000;
joint={0.1 0.2 0.2,
       0.1 0.1 0.3};
call dgsample(joint,1000);

proc dgsample(joint,times);
    local a,b,ma,mb,mjoint,i,result;
    a=1; b=1;  @ Beginning from (1,1). @
    ma=zeros(times,1); mb=zeros(times,1);
    mjoint=zeros(rows(joint),cols(joint));
    i=1;
    do while i<=times;
        a=usample(seqa(1,1,rows(joint)),joint[.,b]/sumc(joint[.,b]));
        b=usample(seqa(1,1,cols(joint)),joint[a,.]'/sumc(joint[a,.]'));
        ma[i]=a;
        mb[i]=b;
        mjoint[a,b]=mjoint[a,b]+1;
        i=i+1;
    endo;
    result=mjoint/sumc(sumc(mjoint));
    print joint;
    print result;
    retp(result);
endp;

proc usample(x,p);
    local n,u,y;
    n=rows(x);
```

```

if n>10;
    errorlog "ERROR: This function can sample only n<=10.";
    retp(".");
endif;
if rows(x)/=rows(p);
    errorlog "ERROR: Rows of x and its p must be the same.";
    retp(".");
endif;
_fcmptol=1e-2;
if fne(sumc(p),1);
    errorlog "ERROR: Sum of the probabilities must be 1.";
    retp(".");
endif;
u=rndu(1,1);
if u>=0 and u<p[1];
    y=x[1];
elseif u>=p[1] and u<(p[1]+p[2]);
    y=x[2];
elseif u>=p[2] and u<(p[1]+p[2]+p[3]);
    if rows(x)<3;
        y=x[2];
    else;
        y=x[3];
    endif;
elseif u>=p[3] and u<(p[1]+p[2]+p[3]+p[4]);
    if rows(x)<4;
        y=x[3];
    else;
        y=x[4];
    endif;
elseif u>=p[4] and u<(p[1]+p[2]+p[3]+p[4]+p[5]);
    if rows(x)<5;
        y=x[4];
    else;
        y=x[5];
    endif;

```

```

elseif u>=p[2] and u<(p[1]+p[2]+p[3]+p[4]+p[5]+p[6]);
    if rows(x)<6;
        y=x[5];
    else;
        y=x[6];
    endif;
elseif u>=p[2] and u<(p[1]+p[2]+p[3]+p[4]+p[5]+p[6]+p[7]);
    if rows(x)<7;
        y=x[6];
    else;
        y=x[7];
    endif;
elseif u>=p[2] and u<(p[1]+p[2]+p[3]+p[4]+p[5]+p[6]+p[7]+p[8]);
    if rows(x)<8;
        y=x[7];
    else;
        y=x[8];
    endif;
elseif u>=p[2] and u<(p[1]+p[2]+p[3]+p[4]+p[5]+p[6]+p[7]+p[8]+p[9]);
    if rows(x)<9;
        y=x[8];
    else;
        y=x[9];
    endif;
else;
    y=x[10];
endif;
retp(y);
endp;

```

画面表示

|             |            |            |
|-------------|------------|------------|
| 0.10000000  | 0.20000000 | 0.20000000 |
| 0.10000000  | 0.10000000 | 0.30000000 |
| 0.10400000  | 0.19600000 | 0.21100000 |
| 0.096000000 | 0.10500000 | 0.28800000 |

最初に示した離散 Joint 分布の値に近い割合で出てくることが上の結果からわかると思います。これは、それぞれのステージで Gibbs Sampling で得られた a および b を使って、 $mjoint[a,b]=mjoint[a,b]+1$ ; によって、その配列(a,b)のところを 1 足して勘定していき、最終的に、 $result=mjoint/sumc(sumc(mjoint))$ ; で列と行の合計の分母分の  $2 \times 3$  の行列の要素のそれぞれの頻度を分子にもって行くことで、合計が 1 の頻度の割合が計算されます。プログラムのには、GAUSS にはある確率でサンプリングする関数は存在しないのでモンテカルロシミュレーションの最初のところでやったように一様分布  $U[0,1)$  を用いて、その区間を確率分布の割合であらかじめ区切っておいて、その一様乱数の出方に応じてその区間に相当する目の出方を計算してやります。それが、usample の procedure です。これは 10 個のサンプルまで合計が 1 になる離散分布にもとづいて引くことができます。

技術的には、procedure usample(x,p)において、インプットの x と p は列で与えられるのですが、エラーチェックを冒頭でやってやります。x の行数 (すなわちサンプルの個数) n が 10 よりも大きいときは想定していないので If 文で除外します。同様に、x と p の行数が一致していない設定、それに、p の合計が 1 におおよそのところならない設定も除外してやります。なお、「おおよそ」1 にならないかどうかを判別するために、fne(a,b) というファジー組込み関数を用いて、その前にそのグローバル設定で 1e-2 で 0.01 に許容桁数を設定してやります。これは 0.3333... が 3 つの場合など必ずしも合計が 1 にならないものにも対応しています。メインのアルゴリズムは、0 から 1 までの区間を確率で区切ってやって、その中にあればどれを選択するのかを決めています。ただし、確率の最終合計が 1 よりも小さい場合、次の elseif に行ってしまうので、これを防止するために、

```
if rows(x)<3;
    y=x[2];
else;
    y=x[3];
endif;
```

などというように、サンプルの個数がそのステージの区切り番号よりも小さい場合には、1 つ戻ってサンプルのリターン y を設定しています。そうでない通常の場合は、else 以降に行って一番大きいインデックスのサンプルが y に設定されます。

上の離散 Gibbs Sampling のケースは、Joint 分布の確率行列が  $2 \times 3$  の場合ですが、同じプログラムで冒頭の行列を変更することによって、最大  $10 \times 10$  まで計算できます。またパラメータ c を加えて、式を一本加えて、3 本の式を a,b,c を代入していくものにすれば、3 者によるような 3 次元のケースにも簡単なプログラムの変更によって対応できます。これらは、ゲーム戦略などの行列にもすぐに実用できる分野です。

次に、今度は離散確率分布ではなく一般的なケースを考えます。今度は、上の行列のインデックスである a および b に相当するところを平均  $\mu$  と Var の  $\sigma^2$  として、その初期値をそれぞれ 0 として Gibbs Sampling で逐次的にシミュレーションします。初期値  $\mu$  と  $\sigma^2$  を

それぞれ 0 に止めておいて、上付きの(k)を第 k 番目のステージであるとする、

$$\mu^{(k)} \sim N(\bar{x}, \sigma^{2(k-1)}/n)$$

$$\sigma^{2(k)} \sim (\sum (x - \mu^{(k)})^2) / (df=n)$$

という Gibbs Sampling を繰り返します。ただし、 $\bar{x}$  のところには、元の X のデータを与えて、その平均を  $\bar{x}$  とするものとします。第 1 ステージでは  $\sigma^{2(0)}$  は 0 ですから  $\mu^{(1)}$  は  $\bar{x}$  そのものとなります。また上の過程の 2 つ目の式は、もともと与えるデータの個数を n とした自由度の  $\sigma^2$  に対する分散ターム  $(x - \mu)^2$  でもってその何倍になるかで、直感的には、ノーマライズされたものに対する割合でサンプル平均の分散を得ます。なお、1 つめの式の分散はサンプル平均の分散ですから、当然のことながら、n で割ったものになります。これを GAUSS の書式にすれば、

```
myu=sqrt(var/n)*rndn(1,1)+xbar;
var=sumc((x-myu)^2)/rndx2(1,1,n);
```

となります。初期値 my と var をそれぞれ 0 としておいて、x ベクトルの分布をあらかじめ決めておいて、そこから  $\bar{x}$  を x ベクトルの平均としてやって、そこから出発させ times 回のステージを繰り返します。

プログラム

```
new; cls;
rndseed 1000;
x=rndnorm(20,1,5,2);          /* mean=5  sigma=2 (var=4) */
call gsample(x,0,0,10000);    /* myu0=0 var0=0 times=1000 */
```

```
proc(2)=gsample(x,myu0,var0,times);
local myu,var,n,xbar,myum,varm,i;
myu=myu0; var=var0;
n=rows(x);
xbar=meanc(x);
myum=zeros(times,1); varm=zeros(times,1);
i=1;
do while i<=times;
    myu=sqrt(var/n)*rndn(1,1)+xbar;
    var=sumc((x-myu)^2)/rndx2(1,1,n);
    myum[i]=myu;
    varm[i]=var;
    i=i+1;
end;
/* Display and Graph the Results */
```

```

print "posterior mean=" meanc(myum);
print "posterior mean of var=" meanc(varm);
library pgraph;
graphset;
pqgwin many;
call hist(myum,50);
call hist(varm,50);
xy(seqa(1,1,times),myum);
xy(seqa(1,1,times),varm);
retp(myum,varm);
endp;

```

```

proc rndx2(r,c,df);
  local m,x,i;
  m=zeros(r,c);
  i=1;
  do while i<=c;
    x=rndn(r,df);
    x=x.*x;
    m[:,i]=sumc(x');
    i=i+1;
  endo;
  retp(m);
endp;

```

```

fn rndnorm(r,c,m,s)=s*rndn(r,c)+m;

```

なお、GAUSS には 2 の乱数関数はありませんから、定義にしたがって標準正規乱数から作成した簡易な乱数関数を用いています。また、標準偏差が  $s$ 、平均が  $m$  の正規乱数は、上のように標準正規乱数に  $s$  をかけて伸ばした上で、 $m$  だけ平行移動させればできます。これらを `procedure` にしてプログラムの末尾に付属させて、組込み関数と同じように使っています。

画面表示

```

posterior mean=          4.9766538
posterior mean of var=    3.3166233

```

### Metropolis-Hastings

この方法は、Proposal をいつも受け入れて次のステージに移行する Gibbs Sampling とは異なり、Proposal Distribution の前のものと当期のものの割合が 0 と 1 の間の一様乱数よりも大きい小さいかによって、受け入れるのか受け入れないのかを判別して、受け入れる場合にのみ、次のステージ（または状態）に移行し、そうではない場合にはその状態にとどまるアルゴリズムです。すなわち、手順は、ある Proposal される Target Distribution があって、そこにある任意の初期状態から出発して（仮に 0 としましょう）、下の例では、 $x$  が 0 から出発して、今の状態の  $x$  すなわち  $x_p$  が  $x$  に下限上限があらかじめ定められた一様乱数の和から得られるものとして、 $N(0,1)$  の標準正規分布の前の状態と今の状態の割合を計算して、それが 0 と 1 の間の一様乱数の値よりも大きければ、その Proposal を受諾して、 $x_p$  を  $x$  に代入することによって次の状態に移行します。これを `times` 回繰り返したものが Metropolis-Hastings のアルゴリズムです。なお、慣習上またコンピュータの桁の実行桁の問題上、通常は、一様乱数と分布の割合の両者の自然対数をとることに比較がなされ判別されます。以下のプログラムでは、自然対数をとった分布の差、つまり分布の割合が計算され、定数部分がキャンセルアウトして、指数関数の部分のマイナスがついている項の差をとっているので、 $\alpha = (-x_p^2 + x^2)/2$  となっています。次の例のように完全な形で書いても、結局は定数項がキャンセルアウトされますから同じことです。

プログラム

```
new; cls;
```

```
rndseed 1000;
```

```
call mhsample(2,0,10000);
```

```
proc(2)=mhsample(up,x0,times);
```

```
    local x,mx,accept,i,xp,alpha,u,ratio;
```

```
    x=x0;
```

```
    mx=zeros(times,1); accept=0;
```

```
    i=1;
```

```
    do while i<=times;
```

```
        xp=x+rndunif(1,1,-up,up);
```

```
        alpha=(-xp^2+x^2)/2; @ Log difference of the target distribution of N(0,1). @
```

```
        u=ln(rndu(1,1));
```

```
        if u<alpha;
```

```
            x=xp;
```

```
            accept=accept+1;
```

```
        endif;
```

```
        mx[i]=x;
```



```

        i=i+1;
    endo;
    ratio=accept/times;
/* Display and Graph the Results */
    print "Acceptance Ratio=" ratio;
    print "mean=" meanc(mx);
    print "s.d.=" stdc(mx);
    library pgraph;
    graphset;
    pqgwin many;
    call hist(mx,50);
    xy(seqa(1,1,times),mx);
    retp(mx,ratio);
endp;

fn rndunif(r,c,low,up)=(up-low)*(rndu(r,c)-0.5)+low+(up-low)/2;

```

画面表示



```

Acceptance Ratio=      0.62770000
mean=      -0.016655616
s.d.=      0.97172303

```

上の結果は、初期の何期間かの切り捨てる部分を全くなしで計算しているので正確ではないが、ほぼ Target Distribution の  $N(0,1)$  に近いものになっている。このことは、以下のグラフからもわかる。

## グラフ表示



今度は、何がしの  $x$  と  $y$  からなる 2 次のマイナス関数の Log Target Distribution を考えて  
これをもとに、 $x$  と  $y$  の Bivariate のケ - スをプログラムしてみる。

プログラム

```
new; cls;
```

```
rndseed 1000;
```

```
fn targetdist(x,y)=-(5*x^2-2*5*x*y+5*y^2);    /* You could change this */
```

```
call mhsample(2,2,0,0,10000);  @ Bivariate Case @
```

```
proc(3)=mhsample(upx,upy,x0,y0,times);
```

```
    local x,y,mx,my,accept,i,xp,yp,alpha,u,ratio;
```

```
    x=x0; y=y0;
```

```
    mx=zeros(times,1); my=zeros(times,1); accept=0;
```

```
    i=1;
```

```
    do while i<=times;
```

```
        xp=x+rndunif(1,1,-upx,upx);
```

```
        yp=y+rndunif(1,1,-upy,upy);
```

```
        alpha=targetdist(xp,yp)-targetdist(x,y); @ Log difference of target distribution @
```

```
        u=ln(rndu(1,1));
```

```
        if u<alpha;
```

```
            x=xp; y=yp;
```

```
            accept=accept+1;
```

```
        endif;
```

```
        mx[i]=x; my[i]=y;
```

```
        i=i+1;
```

```
    endo;
```

```
    ratio=accept/times;
```

```
/* Display and Graph the Results */
```

```
    print "Acceptance Ratio=" ratio;
```

```
    library pgraph;
```

```
    graphset;
```


```
    xy(mx,my);
```

```
    retp(mx,my,ratio);
```

```
endp;
```

```
fn rndunif(r,c,low,up)=(up-low)*(rndu(r,c)-0.5)+low+(up-low)/2;
```

## グラフ表示



与えられた Target 分布のケースは、ほぼ 45 度の角度をもって左上に伸びていきます。なお、iteration の数を減らすともっと短いものに、数を増やすともっと長いものになります。また、Target Distribution の  $-2*5*x*y$  の項の 5 のところを若干増やして、例えば、5.5 にすると、Acceptance Rate が高まります。

これは、 $x$  と  $y$  のまわりの点を候補の中からピックアップしてきて、その density の比を計算して比べていることを意味します。Log difference はもともとの density の比と考えられます。上の 2 つのプログラム例は、Metropolis-Hastings のアルゴリズムのほんの 1 例に過ぎません。いろいろなバージョンがあります。

### Metropolis-within-Gibbs

Gibbs Sampling は理論的にはもちろん Metropolis-Hastings のアルゴリズムの特殊なものと考えられるのだが、これを 2 つ組み合わせて、 $\mu$  および  $\sigma^2$  (または  $\sigma$ ) の導出のアルゴリズムにも Metropolis-Hastings を棄却法を導入したのが Metropolis-within-Gibbs です。平均 0 標準偏差 1 の初期状態から出発して、Target とする Distribution をまず、前の状態の  $\sigma^2$  で止めておいて  $\mu$  の状態が異なる分布の割合をとって、一様乱数と大きさを比べて、大きければ Accept し、次の状態にいき、そうでなければそのままの状態にしておいて、また同じことを今度は  $\mu$  を止めておいて、 $\sigma^2$  の状態が異なる分布の割合をとって、同様な判

別を行なって、状態を進めるのかそのままにするのかを決めます。この2つのサンプリングを `times` 回繰り返したものが下の Metropolis-within-Gibbs です。なお、このプログラムは、 $\mu$  および  $\sigma$  の両方にこのアルゴリズムを施したもので、どちらか片方だけにアルゴリズムを施すこともできます。

プログラム

```
new; cls;
rndseed 1000;
x=rndnorm(20,1,5,4);
call mwgsample(x,0,1,2,1,10000);

proc(4)=mwgsample(x,myu0,sigma0,dmyu,dsigma,times);
  local myu,sigma,myum,sigmam,acceptmyu,acceptsigma,i;
  local myustar,sigmastar,lnalpha,u,ratiomyu,ratiosigma;
  myu=myu0; sigma=sigma0;
  myum=zeros(times,1); sigmam=zeros(times,1);
  acceptmyu=0; acceptsigma=0;
  i=1;
  do while i<=times;
    /* Sampling myu */
    myustar=myu+rndunif(1,1,-dmyu,dmyu);
    lnalpha=posterior(x,myustar,sigma)-posterior(x,myu,sigma);
    u=ln(rndu(1,1));
    if u<lnalpha;
      myu=myustar;
      acceptmyu=acceptmyu+1;
    endif;
    myum[i]=myu;
    /* Sampling sigma */
    sigmastar=sigma+rndunif(1,1,-dsigma,dsigma);
    lnalpha=posterior(x,myu,sigmastar)-posterior(x,myu,sigma);
    u=ln(rndu(1,1));
    if u<lnalpha;
      sigma=sigmastar;
      acceptsigma=acceptsigma+1;
    endif;
    sigmam[i]=sigma;
  enddo;
```

```

        i=i+1;
    endo;
    ratiomyu=acceptmyu/times;
    ratiosigma=acceptsigma/times;
/* Display and Graph the Results */
    print "Acceptance Ratio of   myu=" ratiomyu;
    print "Acceptance Ratio of sigma=" ratiosigma;
    print "posterior mean           =" meanc(myum);
    print "posterior mean of sigma  =" meanc(sigmam);
    library pgraph;
    graphset;
    pqgwin many;
    call hist(myum,50);
    call hist(sigmam,50);
    xy(seqa(1,1,times),myum);
    xy(seqa(1,1,times),sigmam);
    retp(myum,sigmam,ratiomyu,ratiosigma);
endp;

proc posterior(x,myu,sigma);
    local n,lnpost;
    n=rows(x);
    lnpost=-(n+1)*ln(sigma)-sumc((x-myum)^2)/(2*sigma^2);
    retp(lnpost);
endp;

fn rndunif(r,c,low,up)=(up-low)*(rndu(r,c)-0.5)+low+(up-low)/2;
fn rndnorm(r,c,m,s)=s*rndn(r,c)+m;

```

画面表示

```

Acceptance Ratio of   myu=      0.54980000
Acceptance Ratio of sigma=    0.65800000
posterior mean           =      4.9095802
posterior mean of sigma  =      3.6115127

```

## グラフ表示

